


### Lehrbeispiel 1.1: Netzwerk mit drei Zweigen



Bibliotheken aus [2a]

Gleichstrom-Analyse  
 Arbeitspunkt-Analyse  
 (Bias Point)

Bild 1.2: Simulationsergebnisse (Lehrbeispiel 1.1) [Stromzählpfeile manuell gesetzt]

\* Schematics Netlist LB\_1.1 \*

V_UA	3	4	24V
V_UB	3	1	12V
V_UC	3	2	5V
R_R1	4	0	1k
R_R2	1	0	1k
R_R3	2	0	1k

\*\* Analysis setup \*\*

.TEMP 20

.OP

\*\*\* RESUMING LB\_1\_1.cir \*\*\*

.probe

.END

- Aufruf der **Lösungen** (nur bei einer gezeichneten Schaltung) über:

> Enable Bias Voltage Display <


[Werte der Knotenpotentiale]

> Enable Bias Current Display <

[Werte der Zweigströme]

Weitere Informationen (ohne Schaltung): *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.2: Grundstromkreis



Bibliotheken aus [2a]

DC-Analyse  
 DC-Main-Sweep

Bild 1.5: Grundstromkreis zum Lehrbeispiel 1.2

\* Schematics Netlist LB\_1.2 \*

```
V_Uq 1 0 10V
R_Ri 1 A 50
R_Ra A 0 {RL}
.PARAM RL=1k
```

\*\* Analysis setup \*\*

```
.DC LIN PARAM RL 1 1k 1
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_2.cir \*\*\*


```
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- a) Trace → Add Trace      V(A)                      [Verlauf der Spannung über  $R_a$ ]
- b) Trace → Add Trace      I(R\_Ra)                    [Verlauf des Stromes durch  $R_a$ ]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.3: Normierte Funktionsdarstellung



Bibliotheken aus [2a]

DC-Analyse  
 DC-Main-Sweep

Bild 1.5: Simulationsschaltung zum Lehrbeispiel 1.3

\* Schematics Netlist LB\_1.3 \*

```
V_Uq 1 0 10V
R_Ri 1 A 50
R_Ra A 0 {RL}
.PARAM RL=1k
```

\*\* Analysis setup \*\*

```
.DC LIN PARAM RL 1 1k 1
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_3.cir \*\*\*


```
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- |                      |                             |                                       |
|----------------------|-----------------------------|---------------------------------------|
| a) Trace → Add Trace | $V(A)/MAX(V(A))$            | [Normierte Spannung über $R_a$ ]      |
| b) Trace → Add Trace | $I(R\_Ra)/MAX(I(R\_Ra))$    | [Normierter Strom durch $R_a$ ]       |
| c) Trace → Add Trace | $V(A)*I(R\_Ra)$             | [Verlauf der Leistung $P_a$ ]         |
|                      | oder: $V(A)*I(R\_Ra)/500mW$ | [Normierte Leistung $P_a/P_{a,max}$ ] |

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.4: Brückenabgleich



Bibliotheken aus [2a]

DC-Analyse  
 DC-Main-Sweep  
 DC-Nested-Sweep

Bild 1.12: Schaltung mit gesetzten Parametern zur Simulation des Brückenabgleiches

\* Schematics Netlist LB\_1.4 \*

```
V_Uq A 0 10V
R_Rx A C 500
R_RG C D 100
R_RN C 0 {RMB}
RT_RS 0 D {(1k*(1-{SS}))+.001}
RB_RS D A {(1k*{SS})+.001}
.PARAM SS=1 RMB=100
```

\*\* Analysis setup \*\*

```
.DC LIN PARAM SS 0 1 0.01
+ PARAM RMB LIST 125 250 500 1500
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_4.cir \*\*\*

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace I(R\_RG)

[Strom durch den Querschnitt der Brücke]

Weitere Informationen: Analysis → Examine Output (Output-File)

### Lehrbeispiel 1.5: Belasteter Spannungsteiler



Bibliotheken aus [2a]

DC-Analyse  
 DC-Main-Sweep  
 DC-Nested-Sweep

Bild 1.15: Simulation eines belasteten Spannungsteilers

\* Schematics Netlist LB\_1.5 \*

```
V_Uq 1 0 10V
R_Ra A 0 {RL}
RT_RS 1 A {(1k*(1-{SS}))+.001}
RB_RS A 0 {(1k*{SS}))+.001}
.PARAM SS=1 RL=100
```

\*\* Analysis setup \*\*

```
.DC LIN PARAM SS 0 1 0.01
+PARAM RL LIST 10 100 1k 10k
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_5.cir \*\*\*

```
.probe
.END
```


• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- a) Trace → Add Trace V(A) [Verlauf der Ausgangsspannung]
- b) Trace → Add Trace I(RB\_RS) [Verlauf des Querstromes]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



### Lehrbeispiel 1.6: Reihenschaltung RL



Bibliotheken aus [2a]  
Transienten-Analyse

Bild 1.22: Schaltung zum Lehrbeispiel 1.6

\* Schematics Netlist LB\_1.6 \*

```
V_Uq 1 0 AC 1V SIN 0 1V 1kHz 0 0 0
R_R1 1 2 50
L_L1 2 0 10mH
```

\*\* Analysis setup \*\*

```
.TRAN 0 5ms 0 1us
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_6.cir \*\*\*


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(1) V(1)– V(2) V(2) [Zeitfunktionen aller Spannungen]

Weitere Informationen: Analysis → Examine Output (Output-File)

### Lehrbeispiel 1.7: Überlagerung von Zeitfunktionen



Bibliotheken aus [2a]  
 Transienten-Analyse

Bild 1.25: Überlagerung im Lehrbeispiel 1.7

\* Schematics Netlist LB\_1.7 \*

```
V_U1 1 0 SIN 0 5V 1kHz 0 0 0
V_U2 2 1 SIN 0 3V 2kHz 0 0 0
V_U3 3 2 SIN 0 1V 3kHz 0 0 0
V_U4 4 3 SIN 0 4V 4kHz 0 0 0
R_R1 4 0 1k
```

\*\* Analysis setup \*\*

```
.TRAN 0 3ms 0 1us
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_7.cir \*\*\*


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

*Trace* → *Add Trace*      V(4)      [Zeitfunktion der Spannung  $u_{R1}$ ]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.8: Einsatz von AC-Messgeräten



Bibliotheken aus [2a]  
 AC-Analyse (feste Frequenz)  
 Einsatz von AC-Metern

Bild 1.29: Schaltung zum Lehrbeispiel 1.8

\* Schematics Netlist LB\_1.8 \*

```
V_Uq      X      0      AC      1V      SIN 0 1V 1kHz 0 0 0
R_R1      Y      Z      50
L_L1      0      Z      10mH
V_Ameter1 X      Y      AC      0
.WATCH    AC      IM(V_Ameter1)  IP(V_Ameter1)
.PRINT    AC      IM(V_Ameter1)  IP(V_Ameter1)
.WATCH    AC      VM([Z],[0])    VP([Z],[0])
.PRINT    AC      VM([Z],[0])    VP([Z],[0])
.WATCH    AC      VM([Y],[Z])    VP([Y],[Z])
.PRINT    AC      VM([Y],[Z])    VP([Y],[Z])
```

\*\* Analysis setup \*\*

```
.AC LIN 1 1k 1k
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1\_8.cir \*\*\*

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

> Watch < [bei: Simulation Status Window] [Anzeige der Messwerte]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



### Lehrbeispiel 1.9: RC-Tiefpass



Bibliotheken aus [2a]

AC-Analyse  
AC-Sweep (Decade)

Bild 1.32: Schaltung zum Lehrbeispiel 1.9

\* Schematics Netlist LB\_1.9 \*

```
V_Uq ein 0 AC 1V SIN 0 1V 1k 0 0 0
R_R1 ein aus 1k
C_C1 aus 0 159.2nF
```

\*\* Analysis setup \*\*

```
.AC DEC 1000 100Hz 100kHz
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1.9.cir \*\*\*


```
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- a) *Trace* → *Add Trace*                      V(aus)              [Amplitudenfrequenzgang der Ausgangsspannung]
- b) *Trace* → *Add Trace*                      P(V(aus))            [Phasenfrequenzgang der Ausgangsspannung]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.10: Reihenschwingkreis



Bibliotheken aus [2a]

AC-Analyse  
 AC-Sweep (Decade)  
 Parametric-Sweep

Bild 1.37: Reihenschwingkreis mit AC-Sweep und Parametric-Sweep

\* Schematics Netlist LB\_1.10 \*

```
V_Uq ein 0 AC 1V SIN 0 1V 1k 0 0 0
C_C1 ein CL 100nF
L_L1 CL aus 40mH
R_R1 0 aus {RV}
.PARAM RV=10
```

\*\* Analysis setup \*\*

```
.AC DEC 1000 100Hz 100kHz
.STEP PARAM RV LIST 10 40 100
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1.10.cir \*\*\*


```
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- |                      |           |   |
|----------------------|-----------|---|
| a) Trace → Add Trace | V(aus)    | [Amplitudenfrequenzgang der Ausgangsspannung] |
| b) Trace → Add Trace | P(V(aus)) | [Phasenfrequenzgang der Ausgangsspannung]     |

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.11: Dreiphasensystem mit unsymmetrischer Sternschaltung



Bibliotheken aus [2a]  
 (V3Phase wurde durch drei  
 Quellen VSIN ersetzt).  
 Transienten-Analyse

Bild 1.48: Simulation der Lastströme

Für die Widerstände gelten in diesem Beispiel folgende Werte:  $R_1 = 120 \Omega$ ,  $R_2 = 200 \Omega$  und  $R_3 = 150 \Omega$ .

\* Schematics Netlist LB\_1.11 \*

```
V_U1M      L1      0      SIN 0 325V 50 0 0 0
V_U2M      L2      0      SIN 0 325V 50 0 0 -120
V_U3M      L3      0      SIN 0 325V 50 0 0 120
R_R1       L1      1      120
R_R2       L2      1      200
R_R3       L3      1      150
R_RH       0       1      1u
```

\*\* Analysis setup \*\*

```
.TRAN      0      40ms 0      10us
.TEMP      20
.OP
```

\*\*\* RESUMING LB\_1\_11.cir \*\*\*


```
.probe
.END
```

• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace I(R\_R1) I(R\_R2) I(R\_R3) I(R\_RH) [Zeitfunktionen  $i_{Lx}$ ]

Weitere Informationen: Analysis → Examine Output (Output-File)

### Lehrbeispiel 1.12: Auf- und Entladen eines Kondensators



Bibliotheken aus [2a]

Transienten-Analyse  
 (in der Netzliste zur Übung  
 mit **Sw\_tClose** und **Sw\_tOpen**)

Bild 1.61: Testschaltung für Sw\_perChange / Netzliste mit Sw\_tClose/Open

\* Schematics Netlist LB\_1.12 \*

```
V_Uq      1 0      10V
R_R1      2 3      10k
C_C1      3 0      10uF  IC=0V
X_U1      2 0      Sw_tClose PARAMS: tClose=500ms ttran=1u Rclosed=1m
+ Ropen=1G
X_U2      1 2      Sw_tOpen PARAMS: tOpen=500ms ttran=1u Rclosed=1m
+ Ropen=1G
```

\*\* Analysis setup \*\*

```
.TRAN      0      1s      0      1us
.TEMP      20
.OP
```

\*\*\* RESUMING LB\_1\_12.cir \*\*\*

```
.probe
.END
```


- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

*Trace* → *Add Trace*                    V(3)                    [Zeitfunktionen beim Auf- und Entladen]

Anmerkung: Die Berechnung dauert hier etwas länger (Step Ceiling = 1us).

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)

### Lehrbeispiel 1.13: Periodisches Umschalten eines vorgeladenen Kondensators



Bibliotheken aus [2a]

Transienten-Analyse

Bild 1.66: Umschalten eines vorgeladenen Kondensators (Anfangszustand  $U_{C0} = 0 \text{ V}$ )

\* Schematics Netlist LB\_1.13 \*

```
V_UP      1      0      DC      0      AC      0
+ PULSE   0V     1V     200us  0us    0us    500us  1ms
R_R1      1      2      500
C_C1      2      0      0.5uF  IC=0V
```

\*\* Analysis setup \*\*

```
.TRAN      0      5ms    0      1us
.TEMP      20
.OP
```

\*\*\* RESUMING LB\_1\_13.cir \*\*\*


```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

Trace → Add Trace V(1) und: V(2) [Zeitfunktionen  $u_p$  und  $u_{C1}$ ]

Weitere Informationen: Analysis → Examine Output (Output-File)

### Lehrbeispiel 1.14: Ladungsausgleich in einer Reihenschaltung



Bibliotheken aus [2a]

Transienten-Analyse  
 (in der Netzliste zur Übung mit **Sw\_tClose**)  
 RH im Bild ausgeblendet

Bild 1.74: Schaltung zur Simulation des Ladungsausgleichs (Lehrbeispiel 1.14)

\* Schematics Netlist LB\_1.14 \*

```
V_Uq      1      0      20V
R_RH      A      0      10Meg
R_R1      0      B      10k
C_CX      D      A      1uF   IC=10V
C_CY      A      B      4uF   IC=20V
X_U1      1      D      Sw_tClose
+ PARAMS: tClose=1ms tran=1us Rclosed=1m Ropen=1G
```

\*\* Analysis setup \*\*

```
.TRAN      0      30ms  0      1us
.TEMP      20
.OP
```

\*\*\* RESUMING LB\_1.14.cir \*\*\*

```
.probe
.END
```


• Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

- |                      |           |                              |
|----------------------|-----------|------------------------------|
| a) Trace → Add Trace | V(D)–V(A) | [Spannungsverlauf $u_x$ ]    |
| b) Trace → Add Trace | V(A)–V(B) | [Spannungsverlauf $u_y$ ]    |
| c) Trace → Add Trace | V(B)      | [Spannungsverlauf $u_{R1}$ ] |

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)



### Lehrbeispiel 1.15: Freie gedämpfte Schwingung



Bibliotheken aus [2a]

Transienten-Analyse

Bild 1.82: Schaltung zur Simulation eines geschalteten Parallelschwingkreises

\* Schematics Netlist LB\_1.15 \*

```
I_IP 0 1 PULSE 0 100mA 10ms 0 0 100ms 200ms
R_R2 1 0 500
C_C1 1 0 30uF
L_L1 1 0 20mH
```

\*\* Analysis setup \*\*

```
.TRAN 0 600ms 0 100us
.TEMP 20
.OP
```

\*\*\* RESUMING LB\_1.15.cir \*\*\*

```
.probe
.END
```

- Aufruf von Variablen und Darstellung von Funktionen im **PROBE-Fenster** über:

*Trace* → *Add Trace*      V(1)      [Schwingungsverlauf des PSK]

Weitere Informationen: *Analysis* → *Examine Output* (Output-File)